JACG

NEWSLETTER
Vol.4 No.11
July 1985
Single Copy
Price $1.50

THE JERSEY ATARI COMPUTER GROUP

884-1642 JACG HOTLINE 884-1642

From the Editor's Desk...

Putting together the JACG newsletter is no small task. It requires the collection of materials from contributors including articles, photos, cartoons and advertising. The material has to be read and sometimes revised. In the interest of uniformity almost everything is reprinted to conform to a standard format. Most material then has to be reduced to about 75% of its original size to optimize paper space without sacrificing readability. Photos and cartoons have to be hauled to the photostat camera and be shot as needed. In addition, there is an editorial to write, trading post and table of contents to produce, and an occasional article to be written just for fun or necessity.

Once all of this is assembled everything has to be laid out in a dummy of the finished product to produce, hopefully, a balanced result. Then comes the physical cut and paste operation of clipping material and gluing it neatly to the layout sheets in the correct order. Last minute emergencies or space requirements sometimes complicate this step.

The point of this report is not to complain or ask for a raise but to let you know that it takes an average of 15 to 20 hours a month to get it all together. I am pleased to tell you that I now have help with this task. Matt Hetman has come forward and volunteered his time to do the mechanicals. This will relieve me of considerable work and may allow me to continue as editor. We all thank Matt in advance for his commitment.

Now, your part. For the next issue we need your copy by JULY 16th. Matt will be putting the whole thing together by himself since I will be away. Be considerate and get material to us on schedule. We'll appreciate it and you'll get a better newsletter.

Frank Pazel

Frank Pazel
Editor-in-Chief, JACG Newsletter

In This Issue

August Article Deadline

July 16th!

Please Submit Your Contribution Early

MARK YOUR CALENDARS!!

JACG Meeting Schedule

August 10, 1985
September 14, 1985
October 12, 1985
November 9, 1985
December 14, 1985

ESCHER SKETCH

by Kirk McDonald - JACG

Did you know that there are 17 different kinds of wallpaper? That is, the mathematicians tell us that there are 17 ways an image can be repeated regularly to fill up a blank wall. This peculiar fact has been utilized by artists to produce intriguing decoration, and at best, compelling works of graphic art. The most famous examples are the drawings of M.C. Escher. (See "The World of M.C. Escher" published by Abrams, Inc. (1971) for an extensive collection of his work.)

You can try your hand at designing all 17 kinds of wallpaper with the program ESCHER SKETCH which runs on a 48K Atari computer. Use a joystick or Koala Pad to sketch an image and the computer will replicate it 6 or 8 times on your TV screen. Then you can save it to disk in Micropainter format, and/or dump it to an Epson FX-80 printer. The disk file can then be read in by Micropainter or MicroIllustrator if you want to add color.

The program is available on a JACG library diskette in either binary form, ESCHER.COM, or as an Action! text file ESCHER.ACT. An instruction manual is included on the file ESCHER.DOC. The program is based on an article by Edward H. Carlson which appeared in the May 1985 issue of Creative Computing magazine. (Another program based on the imagery of Escher appeared in the August 1983 issue of Antic magazine.)



Symmetry 1 - Prairie Dogs

The figures show my attempts to draw interlocking animal images in the 17 possible ways. Symmetry H yielded only rather crystalline patterns--perhaps you will be more successful. I used symmetries 8 and G to illustrate another amusing aspect of the program: patterns which could be implemented with floor tiles. The fascinating subject of "tessalation" has been explored by Martin Gardner in the July and August 1975 issues of Scientific American.

If you develop a more technical interest in why there are 17 kinds of wallpaper you might peruse the book "Symmetry in Art and Science" by Shubnikov and Koptsik (Plenum Press, 1974).

In any case, the program is quite easy to use, and certainly does not require deep understanding of the geometry it is based on. No great art has ever been created by a mathematician!



Symmetry 2 - Dinos



Symmetry 3 - Mutt and Jeff

Symmetry 4 - Ducklings



Symmetry 5 - E.T.



Symmetry 6 - Double Headers



Symmetry 7 - Bee People



Symmetry 8 - Cairo Tiles



Symmetry 9 - Wombats



Symmetry A - Horsemen



Symmetry D - Ghosts



Symmetry B - Coneheads



Symmetry E - Two Faces



Symmetry C - Owls



Symmetry F - China Dolls



Symmetry G - Alhambra Tiles



Symmetry H - Snowflakes

FASCINATING 153

by Kenneth J. Pietrucha - JACG

One of the interesting things about playing with numbers is the things that are discovered along the way. None of the things I am about to tell you happened "One day as I sat down at my computer", or over any special time period, they just happened.

Take for example, the first time I discovered Narsistic or Armstrong numbers. Briefly, these are numbers where the sum of the cubes of the individual digits is equal to the original number. As an example, take the number 153, the first of the Armstrong numbers, and the star of the show. If we take 1 cubed plus 5 cubed plus 3 cubed, we get 1 + 125 + 27 = 153, which is our original number. This in itself is not very earth shattering. As a matter of fact, like prime numbers, Armstrong numbers are often used as programming exercises. Since 153 is the first of these numbers, it is easily remembered.

A variation of Narsistic numbers is that if you add the cubes of the digits of any number that is a multiple of three, and keep repeating the process on the answer, you will eventually reach the number 153.

Here is the program I wrote to test this conjecture. If you have a printer, remove the REM statements from the begining of line 5 and line 160.

```
  1 REM MULTIPLES OF THREE EVENTUALLY
  2 REM REACH 153 WHEN SUM OF CUBES
  3 REM ARE CHAINED.
  4 REM KENNETH J. PIETRUCHA--5/28/85
  5 REM POKE 838,166:POKE 839,238:REM SEND
    RUN TO PRINTER
  9 DIM A$(5)
 10 FOR T=100 TO 1000
 20 IF T/3<>INT(T/3) THEN GOTO 150
 25 LET X=T
 30 A$=STR$(X)
 40 LA=LEN(A$)
 50 FOR N=1 TO LA
 60 B=VAL(A$(N,N))
 70 C=B*B*B
 80 SUM=SUM+C
 90  NEXT N
100 PRINT SUM;" ";:Z=Z+1
110 IF SUM=153 THEN GOSUB 200:GOTO 150
120 X=SUM
130 SUM=0:C=0:B=0
140 GOTO 30
150 NEXT T
160 REM POKE 838,163:POKE 839,246:REM SEND
    RUN BACK TO SCREEN
200 PRINT
201 PRINT T;" = 153 IN ";Z;" ITERATIONS"
202 Z=0:SUM=0:C=0:B=0
205 PRINT :PRINT
210 RETURN
```

Some where along the way, it was pointed out that 5! + 4! + 3! + 2! + 1! = 153. The symbol ! stands for factorial, which represents the product of all the numbers descending from the given number to 1 (ex. 5*4*3*2*1=120). The sum of the factorials from 5! to 1! is equal to 120 + 24 + 6 + 1 = 153. Interesting, isn't it ? Perhaps, but nothing to write home about.

Some one else came along and pointed out the number 153 is equal to the sum of all the numbers between 1 and 17 (a triangular number).

The number 153 is also mentioned in the bible. In John (21:11), Jesus and his disciples went fishing on the Sea of Tiberias. When they hauled in the catch, your guessed it, 153 fish !

Did John know of the pecularities of the number 153 when he wrote the gosples ? Most of the time the bible uses round numbers like 100 or 1000, so if you see a number like 153, you can be sure it has some special significance. Remember, these people didn't have computers, so the solution must be simple.

What about the computer age ? How does the number 153 fit in ? Well, if we take the ASCII sum of the individual digits, we get 49 + 53 +51 which, believe it or not, equals 153.

If anyone knows any other facts concerning the number 153, I would appreciate your sharing them with me.

VOLUME 11 NUMBER 5 MAY 1985 $2.95

Creative Computing

THE #1 MAGAZINE OF COMPUTER APPLICATIONS AND SOFTWARE

IBM PCAT versus Compaq Deskpro

Modem Magic

More Baud For The Buck

IN-DEPTH EVALUATIONS:

- ISM Express
- Juki 6300 Printer
- WordStar 2000 Plus
- MSBasic 2.0 For The Mac

A New Approach To The Knight's Tour

Tutorial: MSBasic 2.0 For The Mac

How a Consulting Engineer Uses Lotus

Try This Escher Sketch Pad

Color Computer Enhanced Keyboard

Columns: Apple, Atari, IBM, Tandy, Industry Insider, New

Escher Sketch Pad

Edward H. Carlson

The computer is a splendid toy for the mind. Play is nature's classroom, teaching adult and child alike about the possible and the impossible in this amazing universe we skate through. I hope you will join me over the next months in playing with the computer and the ideas that it can model for us.

This month we sketch some intriguing doodles. In later columns I will model the population explosion of castaways on a desert isle, help you invent planets in alternate solar systems, analyze the concepts that underlie management games—*M. U.L.E.* and Hammurabi, for example—relating them to economics, and help you build many other computer toys. Please write to me at *Creative Computing* if you have ideas you would like to contribute.

Escher and the Arabs

Among the many fascinating drawings of Maurits Cornelis Escher are some that tile all space with interlocking figures—angels and devils, birds and fish, horsemen, lizards, and various grotesque monsters. The figures not only fill plane space, leaving no gaps and by repetition extend to infinity, but often are related to one another by certain rotation and reflection symmetries.

Islamic artists (forbidden to represent life forms in their art) created many such tilings with purely geometric forms of intricate design. Escher in his travels studied the Arabic decorations of the Alhambra palace in Granada, Spain (the palace in which Isabella and Ferdinand met with Columbus and approved his plan to sail west to the Indies).

Grab pencil and paper and try to sketch a simple Escher-like drawing—maybe a side view of a cat standing with all four feet showing and legs intertwining with those of another cat, upside down and facing the other way. You find yourself struggling to interlock the figures properly. Now load the program in Listing 1 and draw on the screen. Automatically the figures related by symmetry to the one you are drawing appear on the screen so you can detect any gaps and overlaps immediately. Unless you are somewhat artistic and very patient, you may not produce a satisfactory Escher-like interlocking drawing at all.



Figure 1. Escher's "horseman."

OK then, dream you are the architect for a Moorish sultan's harem, and design intricate tile patterns of high symmetry. In either case, borrow or buy a book of Escher's work and analyze the drawings in light of what the computer program shows you about symmetry in the tiling task.

Staring at the Harem Floor

The theory of tiling (tesselation) needs only a few concepts.

An outline, called the *unit cell*, must fill the plane when you lay copies of it side by side—no gaps, no overlaps, and all cells oriented the same way. That is, you slide one cell (without letting it twist) by its own length to get a neighboring cell. Such a sliding is called a *translation*.

"But" you say, "how about the common bathroom tile of octagons and squares—two dissimilar shapes that together tile the plane?" All such composite outlines as well as irregular outlines like Escher's "horseman" (Figure 1) can be included by adding inner structure to the basic outlines: parallelogram, rectangle, square, and hexagon.

The cells rarely show up in the final drawing explicitly. To detect a cell you must reconstruct it by marking repeated elements with a dot and then joining the dots by lines. For example, dot the tip of the front hoof of every white horse in the Horsemen drawing. The infinite set of dots you get is called the *lattice*. Now connect four adjacent dots and you will see the rectangular unit cell. The cell for a given drawing is unique in size and shape, but not in location. If you choose to dot the chin of each black warrior you would get the same set of rectangular cells, but displaced from the "hoof" cells.

Added to the lattice of cells are three kinds of symmetry elements: rotations, reflections, and glides. The requirement that no overlaps or gaps be present in the tiling of the plane can be satisfied in exactly 17 combinations of cell shape and symmetry elements. These are called the 17 plane space groups, and each is named with a special symbol (see lines 2316 through 2348 of Listing 1). In these symbols, p stands for primitive cell (having lattice points at the corners of the cells only) and c for centered, having also a lattice point in the center of each cell. The numbers stand for n-fold rotation axes, m for mirror planes, and g for glide planes.

N-fold axes of rotation are important in many of Escher's drawings. Only values of $n=2,3,4,6$ are allowed by mathematical consistency with the tiling idea. ($N = 1$ means no rotations at all.) By a "two-fold axis" we mean this: Taking an axis perpendicular to the paper, you rotate the drawing by one half a turn. The whole plane of figures will rotate into coincidence with itself.

A two-fold axis does not impose any conditions on the cell shape, but a four-fold axis can be present only when the cell is a square, and then only at the corners or center of the square. Likewise, a three-fold or six-fold axis requires the cell be hexagonal in shape.

Look for reflection planes in Escher's drawings. Holding a pocket mirror perpendicular to the page and moving it around, you may find a spot where the reflection in the mirror exactly matches the part of the drawing you see if you remove the mirror. If so, draw a line along the base of the mirror and you have marked a reflection plane. Planes tiled with oblique cells (non-rectangular parallelograms) cannot have reflection planes.

You can also detect the presence of reflection planes by just eyeballing the drawing. A symmetric figure, such as the front view of a soldier standing at attention, has a reflection plane (line, really) as the midline from head to toe. Nonsymmetric figures that occur in pairs related like the right and left hand of a glove have a reflection plane half way between them if the line joining equivalent points (tip of each thumb) is perpendicular to the reflection plane. (If not, you have a glide plane.)

> **Glide planes are a combination of reflection and translation that have no counterpart in everyday life.**

Glide planes are a combination of reflection and translation that have no counterpart in everyday life. The horsemen drawing of Escher shows the presence of a glide plane in this way: Each horseman is a mirror image in form (related as right glove to left glove) of one facing the other way (of opposite color) but is not correctly placed to be a mirror image. In fact, each horseman is a mirror image "glided" by a half cell height from the mirror position.

Escher colored his tile prints so the viewer could distinguish each figure from its neighbors. Ignore the colors when saying that one figure reflects, rotates, translates, or glides into another.

More about cell shapes: Symmetry does not restrict the proportions of

Listing 1. Escher Sketch Pad.

```
1 GOTO 2000: ================== ESCHER SKETCH PAD ==============================
2 REM file name:ESCHER                   disk name:CC                 E. H. Carlson
100 REM ========================== main loop ====================================
120 ON G GOSUB 410,420,430,440,450,460,470,480,490,500,510,520,530,540,550,560
,570
130 GOSUB 200                                      :REM user draws
140 GOTO  120                                      :REM plot points
200 REM ====================== get dot for screen ==============================
205 X1=X:Y1=Y                                          :REM save dot to erase
210 CH$=INKEY$:IF CH$="" THEN 210                      :REM get keyboard command
211 IF CH$="Q" THEN WIDTH 80:END                       :REM restore screen and end
212 IF CH$=" " THEN 249                                :REM toggle to other draw modes
220 IF CH$="I" THEN Y=Y-1:IF Y<-A2 THEN Y=Y+A :REM up, don't go off screen
221 IF CH$="M" THEN Y=Y+1:IF Y> A2 THEN Y=Y-A :REM down
222 IF CH$="J" THEN X=X-1:IF X<-B2 THEN X=X+B :REM left
223 IF CH$="K" THEN X=X+1:IF X> B2 THEN X=X-B :REM right
240 IF M<3 THEN RETURN                                 :REM plot or erase all sym dots
241 PSET(X1+W2,Y1+H2),0:PSET(X+W2,Y+H2):GOTO 200:REM erase old, move new dot
249 LOCATE 2,2                                         :REM move cursor to upper left
250 M=M+1:IF M>3 THEN M=1                              :REM toggle to other draw modes
251 IF M=1 THEN E=3:PRINT "DRAW ":RETURN               :REM draw  mode
252 IF M=2 THEN E=0:PRINT "ERASE":RETURN               :REM erase mode
253 IF M=3 THEN     PRINT "MOVE ":GOTO 200             :REM move  mode
350 REM ===================== print dot in all cells ===========================
352 PSET(XX+W2,YY+H2),E                       :REM put original     dot on screen
354 X2=XX-A2   :IF XX<0 THEN X2=XX+A2
356 PSET(X2+W2,YY+H2),E                       :REM put left         dot on screen
358 X3=XX-A2+AC:IF XX<0 THEN X3=X3+A
360 Y2=YY-B2   :IF YY<0 THEN Y2=YY+B2:X3=X3-2*AC
362 PSET(X3+W2,Y2+H2),E                       :REM put upper left   dot on screen
364 X4=XX+AC
366 Y2=YY-B2   :IF YY<0 THEN Y2=YY+B2:X4=X4-2*AC
368 PSET(X4+W2,Y2+H2),E:RETURN                :REM put upper center dot on screen
410 REM ---------------------------------------------------------- no symmetry
411 XX= X    :YY= Y    :GOSUB 350:RETURN      :REM plot original points
420 REM -------------------------------------------------------- two-fold axis
421 XX= X    :YY= Y    :GOSUB 350             :REM plot original points
424 XX=-X    :YY=-Y    :GOSUB 350:RETURN      :REM 2-fold rotation
430 REM ------------------------------------------------------ reflection plane
431 XX= X    :YY= Y    :GOBUB 350             :REM plot original points
435 :         YY=-Y    :GOSUB 350:RETURN      :REM reflection
440 REM ----------------------------------------------------------- glide plane
441 XX= X    :YY= Y    :GOSUB 350             :REM plot original points
442 XX= X+A4:YY=-Y     :GOSUB 350:RETURN      :REM glide plane
450 REM --------------------------------------------------------- centered cell
451 XT= X    :YT= Y    :GOSUB 430             :REM reflection plane
452 X = X+A4:Y =-Y+H4:GOSUB 430               :REM glide plane
456 X = XT  :Y= YT               :RETURN      :REM restore X,Y and return
460 REM -------------------------------------------------- 2 reflection planes
461 :                   GOSUB 420             :REM two fold axis implied
462 XX= X    :YY=-Y    :GOSUB 350             :REM first reflection
465 XX=-X    :YY= Y    :GOSUB 350:RETURN      :REM second reflection
470 REM ------------------------------------------------ reflection and glide
471 :                   GOSUB 420             :REM two fold axis implied
472 XX= X    :YY=-Y+H4:GOSUB 350              :REM reflection
478 XX=-X    :YY= Y+H4:GOSUB 350:RETURN       :REM glide
480 REM ------------------------------------------------------- 2 glide planes
481 :                   GOSUB 420             :REM two fold axis implied
482 XX= X+A4:YY=-Y+H4:GOSUB 350               :REM first glide
```

rectangular or oblique cells, nor the angle in oblique cells. In fact, a rectangular or oblique cell could "accidentally" be square in shape. Each cell contains an integer number of complete figures. (A figure often extends outside the boundaries of the cell. If so, there must be an equivalent part extending into the cell from one corresponding figure in another cell.)

The Dancing Dots

The program is written in Basic on an IBM PC. I have been careful to keep the program as free as possible from special features of the IBM, and hope you can adapt it easily to your computer. Apple and Commodore machines, and Atari and Radio Shack machines using Microsoft Basic, can accept this program with only a few lines changed. I have kept the program simple. It uses the medium resolution black and white screen, and the I, J, K, and L keys to move the cursor for drawing. You may want to use color, low or high resolution, arrow cursor keys, or other special features of your own machine.

Lines 400 to 599 of the program are its geometrical heart and can be moved unchanged to any computer. Each of the 17 space groups is generated by a few lines of code giving the x and y coordinates of all equivalent points in a given cell.

In lines 350 to 399, the dots are written on the screen for four complete cells. One cell has its upper left lattice point in the center of the drawing, and from this point you move the dot that draws the figure. For other computers, you may need to change the PSET command to a PLOT or use a LOCATE or similar manner of putting a dot or character on a specified part of the screen.

Lines 200 to 299 let the program user move the dots around on the screen, and erase dots. These lines can in principle be moved to any other computer, but some changes may have to be made, such as changing the INKEY$ construction of the IBM to a GET or INPUT.

For low symmetry cases, the program is moderately fast. You can input about five points per second. But when you are drawing high symmetry diagrams in which the computer must compute the location of up to 48 points for each point you input, and place them on the screen, the pace slows to about one point per second.

Enjoy the program. May all your figures interlock, on the screen and as you skate through life. ■

```
486 XX=-X+A4:YY= Y+H4:GOSUB 350:RETURN      :REM second glide
490 REM ---------------------------------------------------------- centered cell
491 XT= X    :YT= Y    :GOSUB 460           :REM reflection
492 X = X+A4:Y =-Y+H4:GOSUB 460             :REM glide plane
496 X = XT   :Y = YT              :RETURN   :REM restore X,Y and return
500 REM ------------------------------------------------------------- 4-fold axis
501 XX= X    :YY= Y    :GOSUB 350           :REM plot original points
502 XX=-X    :YY=-Y    :GOSUB 350           :REM half turn
504 XX=-Y    :YY= X    :GOSUB 350           :REM quarter turn
506 XX= Y    :YY=-X    :GOSUB 350:RETURN    :REM reverse quarter turn
510 REM ------------------------------------------------------ 4-fold and reflection
511 XT= X              :GOSUB 500           :REM 4-fold axis points
513 X =-X              :GOSUB 500           :REM reflect
514 X = XT                        :RETURN   :REM restore X,Y and return
520 REM ------------------------------------------------------------ 4 fold and glide
521 XT= X    :YT= Y    :GOSUB 500           :REM 4-fold axis points
523 X =-XT+L4                               :REM glide plane
524 Y = YT+L4          :GOSUB 500
525 X = XT   :Y = YT              :RETURN   :REM restore X,Y and return
530 REM ------------------------------------------------------------- 3-fold axis
531 XX= X    :YY= Y    :GOSUB 350           :REM plot original points
533 XX= X*CO+Y*SI                           :REM rotate by 120 degrees
534 YY=-X*SI+Y*CO      :GOSUB 350
535 XX= X*CO-Y*SI                           :REM rotate by -120 degrees
536 YY= X*SI+Y*CO      :GOSUB 350:RETURN
540 REM ------------------------------------------------ 3-fold with reflections
541 XT= X              :GOSUB 530           :REM plot original points
542 X =-X              :GOSUB 530           :REM reflect in vertical plane
545 X = XT                        :RETURN   :REM restore X and return
550 REM ------------------------------------------------ 3-fold with reflections
551 YT= Y              :GOSUB 530           :REM plot original points
552 Y =-Y              :GOSUB 530           :REM reflect in horizontal plane
555 Y = YT                        :RETURN   :REM restore Y and return
560 REM ------------------------------------------------------------------- 6-fold
561 XT= X    :YT= Y    :GOSUB 530           :REM plot original points
562 X =-XT*CO+YT*SI                         :REM rotate by 60 degrees
563 Y =-XT*SI-YT*CO    :GOSUB 530
565 X = XT   :Y = YT              :RETURN   :REM restore X,Y and return
570 REM ------------------------------------------------ 6-fold with reflections
572 YU= Y              :GOSUB 560           :REM plot original points
573 Y =-YU             :GOSUB 560           :REM reflect
575 Y = YU                        :RETURN   :REM restore Y and return
2000 REM ==================== variables and arrays ============================
2110 W =320   :H =200                       :REM screen width and height
2111 W2=W/2   :H2=H/2:W4=W/4:H4=H/4         :REM half screen sizes
2112 R =W*.85:R2=R/2:R4=R/4                 :REM oblique cell size
2113 L =200   :L2=L/2:L4=L/4                :REM cell size
2120 X =0     :Y =0                         :REM start drawing in center of screen
2130 PI=3.141593
2134 C =L4    :D =L2*SIN(PI/3)              :REM triangular cell positions
2135 SI=SIN(PI/3)    :CO=-.5                :REM sin and cos of 120 degrees
2140 M =1     :E =3   :CH$=" "              :REM initially--draw mode
2200 REM ====================== initial screen ================================
2210 CLS:PRINT:PRINT:PRINT                  :REM clear screen
2215 PRINT"       ESCHER SKETCH PAD":                       PRINT:PRINT
2220 PRINT" PRESS CAPS LOCK":                                     PRINT
2222 PRINT" CHOOSE SYMMETRY FROM MENU":                           PRINT
2224 PRINT" I, J, K, M KEYS TO MOVE DOT":                         PRINT
2226 PRINT" SPACE BAR TO TOGGLE FROM DRAW TO NO DRAW MODE":PRINT
2228 PRINT" PRESS Q KEY TO QUIT":                           PRINT:PRINT
2230 PRINT" (PRESS SPACE BAR TO CONTINUE)":                       PRINT
2235 CH$=INKEY$:IF CH$="" THEN 2235
2300 REM ==================== user inital input ==============================
2310 CLS:PRINT:PRINT:PRINT                 :REM clear screen
2314 PRINT "              PICK THE SYMMETRY":PRINT
2316 PRINT " 1  OBLIQUE                                 p1
2318 PRINT " 2  OBLIQUE      2-FOLD AXIS                p211
2320 PRINT " 3  RECTANGULAR REFLECTION PLANE            p1m1
2322 PRINT " 4  RECTANGULAR GLIDE        PLANE          p1g1
2324 PRINT " 5  RECTANGULAR CENTERED & 2 PLANES         c1m1
2326 PRINT " 6  RECTANGULAR TWO REFLECTION PLANES       p2mm
2328 PRINT " 7  RECTANGULAR REFLECTION AND GLIDE        p2mg
2330 PRINT " 8  RECTANGULAR TWO GLIDE PLANES            p2gg
2332 PRINT " 9  RECTANGULAR CENTERED TWO REFLECTION c2mm
2334 PRINT " A  SQUARE       4-FOLD AXIS                p4
2336 PRINT " B  SQUARE       REFLECTION PLANES          p4mm
2338 PRINT " C  SQUARE       REFLECTION AND GLIDE       p4gm
2340 PRINT " D  HEXAGONAL    3-FOLD AXIS                p3
2342 PRINT " E  HEXAGONAL    REFLECTION PLANES          p3m1
2344 PRINT " F  HEXAGONAL    REFLECTION PLANES          p31m
2346 PRINT " G  HEXAGONAL    6-FOLD AXIS                p6
2348 PRINT " H  HEXAGONAL    6-FOLD AND REFLECTIONS     p6mm
2360 Y$=INKEY$:IF Y$="" THEN 2360          :REM user makes choice
2361 G=ASC(Y$)-48:G=G+7*(G 9)              :REM change input to number
2370 A2=R2:B2=H2:A=R:B=H:AC=0:A4=R4        :REM rectangular cells
2371 IF G 3                       THEN AC=20                        :REM oblique
2372 IF G=10 OR G=11 OR G=12 THEN A2=L2:B2=L2:A=L:B=L    :AC=0      :REM square
2373 IF G 12                      THEN A2=L2:B2=D :A=L:B=D*2:AC=L4  :REM triangular
2390 SCREEN 2,0:CLS:WIDTH 40               :REM medium res. color screen, clear
2399 GOTO 100                              :REM begin drawing
9998 REM Other computers: use GET and PLOT for INKEY and PSET
9999 REM Not direct substitutions--see reference manual
```

Cover Contest Winners

U.S.A. $2.50
CANADA $3.00

Antic™

The ATARI® Resource

AUGUST 1983 VOLUME 2, NUMBER 5

- GRAPHICS
- Atari Signs Alan Alda
- 3-D Fuji
- Datasoft's Pat Ketcham
- Escher Sketcher
- 'Maze Maniac' Game

Escher Sketcher

Isometric illusions Isometric illusions

by BENJAMIN BARTELS

The artistic illusions of M.C. Escher are familiar to many. Birds change to fish, water runs uphill, and men climb stairs that seem to be descending. His techniques inspired me to design a program that creates similar isometric improbabilities.

My Escher Sketcher is an isometric sketch pad that uses the joystick to draw boxes and lines in isometric view. When the joystick is moved in the typical four directions, cubes will be drawn on a two-dimensional plane. When the fire button is pressed, the joystick stacks cubes above or below the main plane, giving an illusion of depth. A variety of colors for the blocks is possible, and a "line" mode is included to embellish your designs.

This program uses GTIA Graphics Mode 10. If you do not have a GTIA chip in your ATARI, you will need to install one in order to use this program.

At the start of the program you will be prompted for background color, cursor color, and two different color combinations. The combinations are for the two cube shapes which you can draw; box 1 and box 2. Colors correspond to Table 9.3 on page 50 of your ATARI BASIC Reference Manual. If you press [RETURN] at these prompts, the program will use a set of default colors.

When you have selected all your colors, press [START] to begin drawing. At this point, you will see a flashing cursor which you can move about the screen without drawing. To draw a line with this cursor, push [SELECT]. To move the cursor without drawing, push [SELECT] again.

The [START] button toggles between drawing boxes and moving the cursor. You can use [SELECT] in Box Draw Mode to change the selection of the next box to be drawn between box 1 and box 2.

In either mode, you can use the [OPTION] key to erase the screen and start over.

PROGRAM BREAKDOWN

10-95 Opening program housekeeping.
100-190 User message and screen preparation.
200-290 Position cursor.
300-390 Main program loop.



400-430 Change box colors — note toggle feature.
900-990 Prepare to draw box by setting X and Y coordinates.
1000-1990 Draw top of box.
1199-1150 Subroutines to draw fiigure.
2000-2990 Draw left side.
3000-3990 Draw right side.

```
10 REM *** PRE-SELECT ***
15 DIM CLLR(2,3),D(4)
20 GRAPHICS 0:SETCOLOR 2,0,0:? "ISO-SK
ETCH by BENJAMIN BARTELS"
30 ?:? "COLOR(0-15),LUM(0-14)":?:? "
INPUT BACKGOUND COLOR AND LUM."
32 TRAP 35:BAK=0:INPUT I,J:BAK=I*16+J
35 ? "INPUT CURSOR COLOR AND LUM.":TRA
P 40:CSSR=15:INPUT I,J:CSSR=I*16+J
40 ?:? "BOX #1":? "INPUT TOP COLOR AN
D LUM.":TRAP 50:CLLR(1,1)=6:INPUT I,J:
CLLR(1,1)=I*16+J
50 ? "INPUT RIGHT COLOR AND LUM.":TRAP
 60:CLLR(1,2)=4:INPUT I,J:CLLR(1,2)=I*
16+J
```

continued on page 98

ESCHER SKETCHER *continued from page 55*

```
60 ? "INPUT LEFT COLOR AND LUM.":TRAP 70:CLLR(1,3)=2:INPUT I,J:CLLR(1,3)=I*16+J
70 ?:? "BOX #2"
80 ? "INPUT TOP COLOR AND LUM.":TRAP 90:CLLR(2,1)=236:INPUT I,J:CLLR(2,1)=I*16+J
90 ? "INPUT RIGHT COLOR AND LUM.":TRAP 95:CLLR(2,2)=134:INPUT I,J:CLLR(2,2)=I*16+J
95 ? "INPUT LEFT COLOR AND LUM.":TRAP 100:CLLR(2,3)=66:INPUT I,J:CLLR(2,3)=I*16+J
100 REM *** START POSITION ***
110 ? "START=TOGGLES BETWEEN POSITIONING":? "CURSOR AND DRAWING BOXES"
120 ? "SELECT=TOGGLES BETWEEN BOX #1 AND":? "BOX #2"
130 ? "OPTION=CLEARS SCREEN"
140 ?:? "PRESS START TO BEGIN"
142 IF PEEK(53279)=7 THEN 142
160 X=40:Y=96:GRAPHICS 10:CBX=0
170 POKE 705,CLLR(1,1):POKE 706,CLLR(1,2):POKE 707,CLLR(1,3)
172 POKE 708,CLLR(2,1):POKE 709,CLLR(2,2):POKE 710,CLLR(2,3)
174 POKE 704,BAK:POKE 711,CSSR
200 REM *** POSITION CURSOR ***
205 IF PEEK(53279)<>7 THEN 205
207 D(4)=0:D(3)=0:D(2)=0:D(1)=5:L=0
210 LOCATE X,Y,I
220 COLOR 0:PLOT X,Y:GOSUB 240
225 IF PEEK(53279)=5 THEN GOSUB 280
227 IF PEEK(53279)=3 THEN 160
230 COLOR 7:PLOT X,Y:GOSUB 250:GOTO 220
240 IF PEEK(53279)=6 THEN POP :COLOR I:PLOT X,Y:POKE 764,255:GOTO 1000
245 RETURN
250 ST=STICK(0):IF ST=15 OR ST=10 OR ST=9 OR ST=5 OR ST=6 THEN RETURN
260 J=STRIG(0):IF L=0 THEN COLOR I:PLOT X,Y
262 IF (ST=13 AND J=1) OR ST=7 THEN X=X+1
263 IF (ST=14 AND J=1) OR ST=11 THEN X=X-1
264 IF ST=14 OR ST=7 THEN Y=Y-1
265 IF ST=13 OR ST=11 THEN Y=Y+1
267 X=X+(X=-1)-(X=80):Y=Y+(Y=-1)-(Y=192):POKE 77,0
270 LOCATE X,Y,I:RETURN
280 L=1-L
285 IF PEEK(53279)<>7 THEN 285
290 RETURN
300 REM *** MAIN LOOP ***
310 ST=STICK(0):IF ST=14 OR ST=7 OR ST=13 OR ST=11 THEN 900
320 IF PEEK(53279)=3 THEN 160
330 IF PEEK(53279)=5 THEN 400
335 IF PEEK(53279)=6 THEN 200
340 GOTO 300
400 REM *** CHANGE BOX COLORS ***
410 CBX=1-CBX
420 IF PEEK(53279)<>7 THEN 420
430 GOTO 300
900 REM *** DRAW BOX ***
910 D(4)=D(3):D(3)=D(2):D(2)=D(1)
920 D(1)=1*(ST=14)+2*(ST=7)+3*(ST=13)+4*(ST=11)
922 IF ST=14 AND STRIG(0)=0 THEN D(1)=5
925 IF ST=13 AND STRIG(0)=0 THEN D(1)=6
930 X=X+3*(D(1)=3 OR D(1)=2)-3*(D(1)=1 OR D(1)=4):Y=Y+3*(D(1)=3 OR D(1)=4)-3*(D(1)<3)+10*(D(1)=6)-10*(D(1)=5)
940 IF X<3 THEN X=3
945 IF X>75 THEN X=75
950 IF Y<4 THEN Y=4
955 IF Y>180 THEN Y=180
960 POKE 77,0
1000 REM *** BOX TOP ***
1010 COLOR 1+CBX*3
1020 IF D(1)=6 THEN 2000
1040 IF D(1)=1 AND D(2)=6 THEN GOSUB 1100:GOTO 2000
1050 IF D(1)=2 AND D(2)=6 THEN GOSUB 1500:GOTO 2000
1060 GOSUB 1100:GOSUB 1500:GOTO 2000
1100 REM *** LEFT HALF ***
1110 TRAP 1120:PLOT X,Y:DRAWTO X,Y-4
1120 TRAP 1125:PLOT X-1,Y-1:DRAWTO X-1,Y-3
1125 TRAP 1130:PLOT X-2,Y-2
1130 RETURN
1500 REM *** RIGHT HALF ***
1520 TRAP 1535:PLOT X+1,Y:DRAWTO X+1,Y-4
1535 TRAP 1540:PLOT X+2,Y-1:DRAWTO X+2,Y-3
1540 TRAP 1550:PLOT X+3,Y-2
1550 RETURN
2000 REM *** DRAW LEFT SIDE ***
2010 COLOR 2+CBX*3
2020 IF D(1)=2 THEN 3000
2025 IF D(1)=6 AND D(2)=2 THEN GOSUB 2500:GOSUB 2700:GOTO 3000
2030 IF D(1)>2 THEN GOSUB 2300:GOSUB 2500:GOSUB 2700:GOTO 3000
2040 IF D(2)=2 THEN GOSUB 2300:GOTO 3000
2050 IF D(2)=1 AND D(3)=2 AND D(4)=2 THEN GOSUB 2300:GOSUB 2500:GOTO 3000
2060 GOSUB 2300:GOSUB 2500:GOSUB 2700:GOTO 3000
2300 REM *** TOP WEDGE ***
2305 TRAP 2310:PLOT X,Y+1
2310 TRAP 2320:PLOT X-1,Y+1:PLOT X-1,Y+2
2320 TRAP 2330:PLOT X-1,Y
2330 TRAP 2340:PLOT X-2,Y+1:DRAWTO X-2,Y-1
2340 TRAP 2350:PLOT X-2,Y+2:DRAWTO X-2,Y+3
2350 RETURN
2500 REM *** MID WEDGE ***
```

```
2510 TRAP 2520:PLOT X,Y+2:DRAWTO X,Y+7
2520 TRAP 2530:PLOT X-1,Y+3:DRAWTO X-1
,Y+8
2530 TRAP 2540:PLOT X-2,Y+4:DRAWTO X-2
,Y+8
2540 RETURN
2700 REM *** BOTTOM WEDGE ***
2710 TRAP 2720:PLOT X,Y+8:DRAWTO X,Y+1
0
2720 TRAP 2730:PLOT X-1,Y+9
2730 RETURN
3000 REM *** DRAW RIGHT SIDE ***
3010 COLOR 3+CBX*3
3020 IF D(1)=1 THEN 4000
3025 IF D(1)=6 AND D(2)=1 THEN GOSUB 3
500:GOSUB 3700:GOTO 4000
3030 IF D(1)>2 THEN GOSUB 3300:GOSUB 3
500:GOSUB 3700:GOTO 4000
3040 IF D(2)=1 THEN GOSUB 3300:GOTO 40
00
3050 IF D(2)=2 AND D(3)=1 AND D(4)=1 T
HEN GOSUB 3300:GOSUB 3500:GOTO 4000
3060 GOSUB 3300:GOSUB 3500:GOSUB 3700:
GOTO 4000
3300 REM *** TOP WEDGE ***
3305 TRAP 3310:PLOT X+1,Y+1
3310 TRAP 3320:PLOT X+2,Y+1:PLOT X+2,Y
+2
3320 TRAP 3330:PLOT X+2,Y
3330 TRAP 3340:PLOT X+3,Y+1:DRAWTO X+3
,Y-1
3340 TRAP 3350:PLOT X+3,Y+2:DRAWTO X+3
,Y+3
3350 RETURN
3500 REM *** MID WEDGE ***
3510 TRAP 3520:PLOT X+1,Y+2:DRAWTO X+1
,Y+7
3520 TRAP 3530:PLOT X+2,Y+3:DRAWTO X+2
,Y+8
3530 TRAP 3540:PLOT X+3,Y+4:DRAWTO X+3
,Y+8
3540 RETURN
3700 REM *** BOTTOM WEDGE ***
3710 TRAP 3720:PLOT X+1,Y+8:DRAWTO X+1
,Y+10
3720 TRAP 3730:PLOT X+2,Y+9
3730 RETURN
4000 GOTO 300
```

TYPO TABLE

Variable checksum = 173371

Line num range	Code	Length
10 - 50	FQ	565
60 - 110	JB	560
120 - 207	ZZ	559
210 - 264	ML	470
265 - 340	RA	391
400 - 930	VE	642
940 -1100	SV	387
1110 -2020	HJ	385
2025 -2330	GY	507
2340 -3000	JI	384
3010 -3320	MG	513
3330 -3730	SJ	444
4000 -4000	MD	13

Nonperiodic Tiling

Kirk T. McDonald
Joseph Henry Laboratories, Princeton University, Princeton, NJ 08544
(January 27, 2024)

The article "Escher Sketch" [1] considered only periodic tilings of a plane with a single type of tile.[1] Already in 1976, Penrose had demonstrated nonperiodic tilings, based on 2 different tiles [2, 3, 4]. Recently, nonperiodic tilings with only a single (14-sided) tile have been demonstrated [5, 6].





References

[1] K. McDonald, *Escher Sketch*, JACG Newsletter **4**(11), 18 (1985), http://kirkmcd.princeton.edu/examples/escher.pdf

[2] R. Penrose, *Set of Tiles for Covering a Surface*, US Patent 4,133,152 (filed Jan. 24, 1976), http://kirkmcd.princeton.edu/examples/patents/penrose_us4133152_76_tiles.pdf

[3] R. Penrose, *Pentaplexity*, Eureka No. 39, 16 (1978), http://kirkmcd.princeton.edu/examples/mechanics/penrose_eureka_39-apr_16_78.pdf

[4] R. Penrose, *Pentaplexity*, Math. Intell. **2** 32 (1979), http://kirkmcd.princeton.edu/examples/mechanics/penrose_mi_2_32_79.pdf

[5] D. Smith *et al.*, *An aperiodic monotile* (May 2023), https://arxiv.org/abs/2303.10798

[6] D. Smith *et al.*, *A chiral aperiodic monotile* (May 2023), https://arxiv.org/abs/2305.17743

[1]Of course, each tile can be partitioned into many subtiles. Several of the illustrations of the 17 basic types of periodic tilings in [1] were shown with 2 subtiles.